```
→  /* some examples arising from one-variable calculus
   - the values of π and e
   - factorization
   - values of some functions
   - graphs of functions
   - graphs of a function with a zero of infinite order
   - substitution and expansion
   - limits
   - differentiation
   - integration
   - some improper integrals
   - Taylor polynomials
   - solving ordinary differential equations
   */
```

```
(%i2) /* π and e */
      float(%pi);
      float(%e);
```

(%o1) 3.141592653589793

(%o2) 2.718281828459045

```
(%i4) /* factorization */
      factor(389423891274087282891 31);
      factor(x^3-7*x+6);
```

(%o3) $179^2\ 38711\ 235177\ 133501853$

(%o4) $(x-2)\,(x-1)\,(x+3)$

```
(%i7) /* values of some function */
      sin(%pi/4);
      atan(1);
      log(%e^3);
```

(%o5) $\dfrac{1}{\sqrt{2}}$

(%o6) $\dfrac{\pi}{4}$

(%o7) 3

(%i9) /* graphs of functions */
```
wxplot2d([sin(1/x)], [x,-1/10,1/10], [style, [lines, 0.4, 7]]);
wxplot2d([x,-x,x*sin(1/x)], [x,-1/10,1/10], [style, [lines, 0.4]]);
```

plot2d: expression evaluates to non-numeric value somewhere in plotting range.

(%t8) 

(%o8)

plot2d: expression evaluates to non-numeric value somewhere in plotting range.

(%t9) 

(%o9)

```
→ /* This is the graph of the function
f(x) := exp(-1/x) for x>0 else 0,
which has a zero of infinite order at x=0. */
f(x):= if x>0 then exp(-1/x) else 0$
wxplot2d([f(x)], [x,-1,1], [style, [lines, 3, 8]],
    [ylabel, "y=f(x)"], [title, "f(x):= if x>0 then exp(-1/x) else 0"])$
```

(%t11)




```
→ /* substitution and expansion */
f(t):=t^2+t+2;
g(x):=x^3+2*x^2+5*x+7;
expand(f(g(x)));
```

(%o16) $f(t):=t^2+t+2$

(%o17) $g(x):=x^3+2x^2+5x+7$

(%o18) $x^6+4x^5+14x^4+35x^3+55x^2+75x+58$

```
(%i14) /* limits */
limit(sqrt(n+1)-sqrt(n),n,inf);
limit(sin(x)/x,x,0);
limit((1+1/x)^x,x,inf);
```

(%o12) 0

(%o13) 1

(%o14) %e

```
(%i16) /* differentiation */
diff(x*log(x)-x,x);
diff(atan(x),x);
```

(%o15) $\log(x)$

(%o16) $\frac{1}{x^2+1}$

```
(%i19) /* integration */
integrate(sin(x),x,0,%pi);
integrate(log(x),x,1,%e);
integrate(1/(1+x^2),x,1,inf);
```

(%o17) 2

(%o18) 1

(%o19) $\frac{\pi}{4}$

```
(%i24) /* improper integrals */
/* the Dirichlet integral */
integrate(sin(x)/x,x,0,inf);
/* the Fresnel integrals */
integrate(sin(x^2),x,-inf,inf);
integrate(cos(x^2),x,-inf,inf);
/* Fourier transform */
integrate(exp(-2*%pi*%i*x*ξ-%pi*x^2),x,-inf,inf);
integrate(exp(-%i*x*ξ-x^2/2),x,-inf,inf)/sqrt(2*%pi);
```

(%o20) $\frac{\pi}{2}$

(%o21) $\frac{\sqrt{\pi}}{\sqrt{2}}$

(%o22) $\frac{\sqrt{\pi}}{\sqrt{2}}$

(%o23) $\%e^{-\pi \xi^2}$

(%o24) $\%e^{-\frac{\xi^2}{2}}$

```
(%i26) /* Taylor polynomial */
taylor(log(1+x), x, 0, 10);
taylor((1+x)^(1/2), x, 0, 10);
```

(%o25)/T/ $x-\frac{x^2}{2}+\frac{x^3}{3}-\frac{x^4}{4}+\frac{x^5}{5}-\frac{x^6}{6}+\frac{x^7}{7}-\frac{x^8}{8}+\frac{x^9}{9}-\frac{x^{10}}{10}+...$

(%o26)/T/ $1+\frac{x}{2}-\frac{x^2}{8}+\frac{x^3}{16}-\frac{5x^4}{128}+\frac{7x^5}{256}-\frac{21x^6}{1024}+\frac{33x^7}{2048}-\frac{429x^8}{32768}+\frac{715x^9}{65536}-\frac{2431x^{10}}{262144}+...$

(%i29)
```
/* solving a differential equation */
ode2('diff(u,x)=u, u, x);
ode2('diff(u,x)=u^2, u, x);
ode2('diff(u,x,2)+'diff(u,x)+u=sin(x), u, x);
```

(%o27) $u=\%c\,\%e^{x}$

(%o28) $-\frac{1}{u}=x+\%c$

(%o29) $u=\%e^{-\frac{x}{2}}\left(\%k1\,\sin\left(\frac{\sqrt{3}\,x}{2}\right)+\%k2\,\cos\left(\frac{\sqrt{3}\,x}{2}\right)\right)-\cos\left(x\right)$